



September 10 - 12, 2007

Pilsen, Czech Republic

S-Z AC ANALYSIS OF SWITCHED CIRCUITS IN PSPICE

PROF. ING. DALIBOR BIOLEK, CSC.

ING. VIERA BIOLKOVÁ

DOC. DR. ING. ZDENĚK KOLKA

Abstract: *The paper explains basic ideas how to model and analyze the AC behavior of circuits containing analog switches, controlled by periodical external clock, namely Sample-Hold circuits and switched capacitor filters. A procedure of making up the so-called s-z models is described as well as their implementation in PSPICE.*

Key words: *Switched circuit, Sample-Hold, Switched capacitor, AC analysis, PSpice*

INTRODUCTION

In the area of analog signal processing, the term “switched circuit” denotes a circuit, containing analog switches with the possibility of controlling their states electronically. A number of circuits from various applications belong to this group. Let us mention switched capacitor filters, Sample-Hold circuits, peak detectors, mixers, modulators of shift-keying techniques, a wide range of switch-mode power supplies, switched DC-DC converters, etc. The switch state can be controlled either by an independent clock signal (the case of switched capacitor circuits), or it depends on the internal state of the circuit (e.g. the feedback voltage regulator of DC-DC converter). Then we talk about circuits with external and internal switching [1].

Switched circuits belong to systems in which the processes have very dissimilar time scales (the process of switch action on the one hand, and the envelope or average value of the signal on the other hand). During the transient analysis, the program must control the time step according to the fast phenomena, which results in long simulating runs. Another problem is associated with the ineffective process of the time step control when simplified (behavioral) models of idealized switches are used.

The conventional circuit analysis programs do not provide any support for an effective analysis of such a kind of networks. The only tool is represented by the transient analysis, whose time step is determined by the switching frequency. Also, the direct computation of the periodical steady state as well as the AC analysis is not

available (with the exception of special RF simulators which do not target the switched circuits).

In the past, a number of methods have been developed for computer modeling and simulation of switched circuits with negligible ratios of time constants of transient phenomena and lengths of switching phases. These methods were created particularly for the analysis of circuits with capacitors “charged in jumps”, i.e. for switched capacitor filters. The well-known papers by Vlach et al. [1], [2], [3] and his special programs such as SWANN [2] should be mentioned here. The above programs work on quite a different principle than SPICE-compatible programs. They employ ideal models of circuit elements: ideal switches (short connection in the on-state, open circuit in the off-state), capacitors, and ideal transforming cells (controlled sources and ideal operational amplifiers). The circuit equations, which are made up from the switched circuit topology and are subsequently solved by special algorithms, are the so-called z-domain charge equations, where z is the operator of the z-transform.

A certain progress in the computer simulation of switched circuits was achieved during the 1980s also in our country: The first COCOSC and MINICOCOSC programs for the symbolic analysis of idealized switched capacitor filters in the world were developed in cooperation between Technical University of Brno and Military Academy Brno [4-7]. Subsequently, the first program for semisymbolic analysis of switched networks SCSK [8] and the first program for time-domain analysis SCC [9] were developed. The latter provided (on the then

8-bit computers) for the analysis of forced circuit responses to various signals from the library and even for the analysis of the periodical steady state. Another important contribution by the present applicants to the theory and praxis of switched circuit modeling and simulation is the so-called theory of generalized transfer functions (GTF) [10], which enabled the AC analysis of real switched networks.

At that time, as a consequence of the development of switched circuit applications in power electronics, an urgent necessity of computer simulation appeared for circuits which were a sort of opposite to the switched capacitor circuits. The switched DC-DC converter is a typical representative: the time constants of the transients are longer (much longer in the ideal case) than the duration of the switching phases. The methods of modeling switched circuits with negligible time constants cannot be used in such cases. That is why new methods, based on the so-called averaging approach, have been developed [13], [14], which replace the switched circuit, at the cost of certain simplification, by an equivalent model for the signal envelope or its average value. A great advantage consists in the easy integration of the averaged models into the SPICE-compatible programs, but on the assumption that the user is familiar with the technique of averaged modeling. There are also some drawbacks in this approach. The averaged models stop working for phenomena that are fast in comparison with the switching period. That is why the frequency responses of DC-DC converters, used e.g. in the design and optimization of feedback regulators of the output voltage, provide relevant results only up to one half of the switching frequency. Also, there are no known procedures of how to extract information from the averaged model about the side effects, which are related to aliasing in the frequency domain.

The averaging also smoothes all the information about the switching character of the signals. That is why one cannot find, for example, the steady-state voltage ripple from the averaged model. Such characteristics are only available via time-consuming transient analysis.

Our preliminary analyses [15] indicate that some of the above problems should be overcome using the theory of generalized transfer functions, because modeling, based on GTF, sets no limits to the length of transient phenomena in the circuit. That is why the GTFs would offer a more universal methodology for modeling the switching phenomena either in the circuits of power electronics or in the switch capacitor networks.

Let us conclude that the algorithms hitherto developed for an effective analysis of switched circuits, cannot be conventionally implemented in SPICE-compatible programs in most cases. There is no universal method simultaneously applicable for the analysis of switched circuits with small time constants (switched capacitor circuits), and with large time constants (switched DC-DC converters), and thus for the hybrid circuits either (e.g. the Sample-Hold circuits, which combine the extremely small and large time constants). That is why a number of special programs with different internal algorithms exist, which enable the analysis of circuits from limited application groups. The source codes of these programs are not freely available and there is no legal possibility of

modifying and redistributing them. This is in contradiction with our needs for a direct integration of new methods of modeling and analysis into the simulator.

In this paper, we describe a possibility of direct AC analysis of switched circuits in PSpice, without any limitations to the values of time constants in the circuit being analyzed. The mathematical tool which will be used is the GTF approach. Note that we limit our considerations to the AC analysis. The discussion of other types of Spice analyses is beyond the scope of this paper.

1 GENERALIZED TRANSFER FUNCTIONS

Consider a simple model of Sample-Hold (S-H) circuit in Fig. 1 (a) and the corresponding waveforms in Fig. 1 (b). At any moment, the circuit operates in one of two possible phases, denoted in Fig. 1 as phase 1 (the clock signal ϕ is HIGH and the switch is closed), and phase 2 (the clock signal ϕ is LOW and the switch is open). The ratio ε of duration of phases 1 and 2 can be from 0 to 1. In real S-H circuits, ε is below $\frac{1}{2}$, typically $\varepsilon = 0.1$.

Due to nonzero time constant of charging the capacitor through the nonzero switch on-resistance, the output voltage at time instant $kT + \varepsilon T$ of finalizing the sample selection will not exactly equal to the input voltage. Analyzing this dynamical error of sample selection, we concern not in the entire output signal, but only in its discrete values at the end of switching phases 1. The corresponding dots are marked in Fig. 1 (b) by unfilled rings. Dependence of the "quality" of the sample selection on the frequency of sampled signal can be evaluated as follows:

1. We excite the S-H circuit by a harmonic voltage with a certain frequency. This signal will gradually set the circuit in the (quasi)periodic steady state.
2. In the output signal we identify points which we are concern in. In our case, these points lie at ends of switching phases 1.
3. The points from step 2 we interleave with a harmonic signal with the frequency equal to those of the input signal. This task is unambiguous [10]. We obtain a signal called equivalent signal v^{1e} to signal v [10].
4. Comparing two harmonic signals v^{1e} and v yields values of generalized amplitude and phase frequency responses of the circuit on the concrete frequency of input signal.
5. We repeat the above steps for various frequencies of excited signal, taking them from required limits on the frequency axis in order to obtain generalized frequency responses.

Note that just the GTF $K(s, z)$ is a compact mathematical description of the generalized frequency response. The frequency responses can be acquired from GTF after well-known substitutions $s = j\omega$ and $z = \exp(j\omega T)$, where T is the switching period.

It stands for reason that frequency responses will depend on how to choose the "points of interest" on the output waveform. As an example, the second equivalent signal v^{2e} is shown in Fig. 1 (b), which is interleaved with crosses at the ends of phase 2. Another point, marked by filled circle T_p seconds after end of phase 1, contains

information about the value of output voltage at time instant of finishing AD conversion by converter, attached to the S-H output. Another equivalent signal and frequency response will be assigned to these points. Comparing it with the signal v^{1e} and the corresponding frequency response, one can deduce an amount of voltage drop of the output voltage during the AD conversion in the HOLD state.

Let us conclude that an infinity set of equivalent signals can be assorted to single signal of switched circuit, and one GTF to any relation of input-equivalent output signal. In this way, one can model the switched circuit from more points of view, depending on circuit features we are interested in.




Fig. 1: (a) Model of the Sample-and-Hold circuit, (b) the corresponding waveforms.

2 EXAMPLE OF GTF EVALUATION FOR S-H CIRCUIT

Consider the model of S-H circuit in Fig. 1 (a) with nonzero and signal independent on-resistance R_{on} and infinity off-resistance of the switch. Then the circuit is described by linear equations in each switching phase. The following equations evaluate output voltage at the ends of given phases. Index of this phase is marked by the superscript.

Phase 1: $kT < t \leq kT + \epsilon T$

$$v^1(kT + \epsilon T) = v^2(kT) e^{-\frac{\epsilon T}{\tau_1}} + \int_0^{\epsilon T} g_1(\xi) w(kT + \epsilon T - \xi) d\xi \quad (1)$$

Phase 2: $kT + \epsilon T < t \leq kT + T$

$$v^2(kT + T) = v^1(kT + \epsilon T) e^{-\frac{(1-\epsilon)T}{\tau_2}}, \quad (2)$$

where

$$\tau_1 = (R_{on} \& R)C, \tau_2 = RC, g_1(t) = \frac{1}{R_{on}C} e^{-\frac{t}{\tau_1}}. \quad (3)$$

Now consider harmonic input signal, described by a complex phasor

$$w(t) = \hat{W} e^{st}, s = j\omega. \quad (4)$$

Left sides of (1) and (2) describe samples of output voltage at the ends of switching phases 1 and 2. In other words, they are samples of equivalent signals $v^{1e}(t)$ and $v^{2e}(t)$ from 1 (b), which nature is also harmonic. Let us describe these signals by complex phasors

$$v^{1e}(t) = \hat{V}^{1e} e^{st}, v^{2e}(t) = \hat{V}^{2e} e^{st}. \quad (5)$$

Substituting (4) and (5) to (1) and (2) and arrangement yield the resulting form of Eqs. (1) and (2):

$$\hat{V}^{1e} = e^{-\frac{\epsilon T}{\tau_1}} \hat{V}^{2e} z^{-\epsilon} + G_1(s, z) \hat{W}, \quad (6)$$

$$\hat{V}^{2e} = e^{-\frac{(1-\epsilon)T}{\tau_2}} \hat{V}^{1e} z^{-(1-\epsilon)}, \quad (7)$$

where

$$z = e^{sT}, \quad (8)$$

$$G_1(s, z) = \int_0^{\epsilon T} g_1(\xi) e^{-s\xi} d\xi = \frac{R}{R + R_{on}} \frac{1 - e^{-\frac{\epsilon T}{\tau_1}} z^{-\epsilon}}{1 + \tau_1 s}. \quad (9)$$

Eqs. (6) and (7) get formulas for phasors of both equivalent signals:

$$\hat{V}^{1e} = K^1(s, z) \hat{W}, \hat{V}^{2e} = K^2(s, z) \hat{W}, \quad (10)$$

where

$$K^1(s, z) = \frac{G_1(s, z)}{1 - e^{-\frac{\epsilon T}{\tau_1}} e^{-\frac{(1-\epsilon)T}{\tau_2}} z^{-1}},$$

$$K^2(s, z) = K^1(s, z) e^{-\frac{(1-\epsilon)T}{\tau_2}} z^{-(1-\epsilon)}$$

are generalized transfer functions of S-H circuit, corresponding to the above equivalent signals $v^{1e}(t)$ and $v^{2e}(t)$.

3 GTF IMPLEMENTATION IN PSpICE

Equations (6) and (7) can be easily implemented in PSpice program for a direct AC analysis of S-H circuits. Below is a complete list of the circuit file for PSpice analysis:

S-H circuit, sampling frequency 100kHz, GTF modeling

```
*
.param C=10nF R=100kOhm Ron=50Ohm fs=100kHz ep=0.1
+Ts={1/fs} RonR={Ron*R/(Ron+R)}
+tau1={C*RonR} tau2={C*R}
+exp1={exp(-ep*Ts/tau1)} exp2={exp((ep-1)*Ts/tau2)}
*****
Vin in 0 AC=1
Eout1 out1 x LAPLACE=
+{V(in)} {R/(R+Ron)*(1-exp1*exp(-s*ep*Ts))/(1+tau1*s)}
Ex x 0 LAPLACE={V(out2)} {exp1*exp(-s*ep*Ts)}
Eout2 out2 0 LAPLACE={V(out1)} {exp2*exp((ep-1)*s*Ts)}
```

```

*****
.step param Ron list 10 20 50 100
.AC dec 100 100 500k
.probe
.end

```

Results of the AC analysis for four different switch on-resistances are in Fig. 2. The resulting frequency responses were carefully checked with the results of steady-state transient analysis of the switched-level model of S-H circuit in Fig. 1 (a). The negligible differences were caused mainly by numerical inaccuracies of expensive transient analysis.




Fig. 2: Amplitude and phase frequency responses of S-H circuit for equivalent signal v^{1e} .

The necessity of analytical derivation of equations (1)-(10) prior their modeling in PSpice is a serious drawback of this method, particularly in the case of more complicated switched circuits. The alternative method described below overcomes this difficulty by means of special numerical computations within the convenient transient analysis.

4 SIMPLIFIED Z-DOMAIN GTF IMPLEMENTATION IN PSpice

The GTF approach truly models the switched circuit behavior also in the frequency region above one half of the switching frequency. However, in many cases only the analysis below Nyquist's frequency is normally performed without the necessity to utilize this unique feature. Then a simplified z-domain modeling can be used instead of the more general s-z description. This

simplification enables easy PSpice modeling, based on the numerical approach, avoiding the expensive analytical preprocessing.

Let us consider that the input signal $w(t)$ remains constant within the individual switching phases. Then equations (1) and (2) can be written as follows:

$$v^1(kT + \varepsilon T) = a_1 v^2(kT) + b_1 w(kT), \quad (11)$$

$$v^2(kT + T) = a_2 v^1(kT + \varepsilon T) + b_2 w(kT + \varepsilon T). \quad (12)$$

Here the coefficients a_1 , b_1 , a_2 and b_2 depend on the circuit topology in phases 1 and 2 and on the duration of switching phases. It is obvious that $b_2 = 0$ if the switch has an infinite off-resistance. The remaining coefficients can be computed in PSpice during the transient analysis by the following method.

The coefficients from Eq. (11) / (12) will be evaluated via the transient analysis of the circuit, which is simplified for the switching phase 1 / 2, within the length T_1 / T_2 of the simulation run. Here $T_1 = \varepsilon T$ and $T_2 = (1 - \varepsilon)T$.

The "a" coefficient is given by the natural response to the initial voltage 1V of the capacitor for the input voltage to be zero at the end of transient analysis.

The "b" coefficient is equal to the final value of forced response to the input voltage of 1V with zero initial condition.

Knowing these coefficients, the frequency response will be computed from the set of z-domain equations

$$\hat{V}^{1e} = (a_1 \hat{V}^{2e} + b_1 \hat{W}) z^{-\varepsilon}, \quad (13)$$

$$\hat{V}^{2e} = (a_2 \hat{V}^{1e} + b_2 \hat{W}) z^{-(1-\varepsilon)}. \quad (14)$$

However, there are two fundamental problems when implementing the above procedures in PSpice:

1 Repeated transient analysis is required in order to obtain results of these analyses for the subsequent analysis. OrCad PSpice does not allow this.

2 After the transient analysis, the AC analysis should be executed, processing the data from the transient analysis. However, PSpice does not support such data sharing.

Problem 1 can be overcome by executing one transient run of length $T_2 > T_1$. The auxiliary models for phases 1 and 2 are analyzed simultaneously. The a_1 and b_1 coefficients are evaluated at time point T_1 , the a_2 and b_2 coefficients at the end of analysis run. To keep the values of all the coefficients at the end of the analysis, we must complete this scheme by ideal Track-Hold blocks with the sampling action at time T_1 . The voltage at their outputs will then correspond to values a_1 and b_1 . To ensure a high precision of the sampling at time T_1 , we use a special BREAK function which is implemented in OrCad PSpice 15.7. Similarly, the ideal Track-Hold block can be modeled, combining the new BREAK and STATE functions for behavioral modeling.

Problem 2 must be solved on the platform of transient analysis. For a particular angular frequency ω , complex equations (13) and (14) can be rewritten as four real equations, utilizing the equality

$$z = e^{j\omega T} = \cos(\omega T) + j \sin(\omega T).$$

The frequency will be defined as a global parameter by the .param statement, with the possibility of stepping it within the required frequency range. The above set of equations must be solved simultaneously with equations

(11) and (12). For a given frequency, the correct value of frequency response is available at the end of the transient run.

The complete PSpice v. 15.7 input file is as follows:

```
SH circuit, z-domain solution
.param C=10nF R=100kOhm Ron=50Ohm fs=1megHz
+ ep=0.1 ep2={1-ep}
+ T1={ep/fs} T2 {(ep2)/fs} T={1/fs} f=1k om={2*pi*f}
.param cos1={cos(om*ep*T)} sin1={sin(om*ep*T)}
+ cos2={cos(om*ep2*T)} sin2={sin(om*ep2*T)}
*****
*precise computation of circuit values for t=0.1us
ex x 0 value={break(0.1u)}
*****
*computing a1=voltage across C at the end of time T1
Xa1 0 outa1 SH1
XTHa1 outa1 a1 TH
.IC V(outa1)=1V
*****
*computing b1=voltage across C at the end of time T1
Vb1 be1 0 1V
Xb1 be1 outb1 SH1
XTHb1 outb1 b1 TH
*****
*computing a2=voltage across C at the end of time T2
Xa2 outa2 SH2
ea2 a2 0 value={V(outa2)}
.IC V(outa2)=1V
*****
*computing AC response via transient analysis!!!
Vwr wr 0 DC=1
Vwi wi 0 DC=0
Ecr1 cr1 0 value={(V(a1)*V(cr2)+V(wr)*V(b1))*cos1+
+(V(a1)*V(ci2)+V(wi)*V(b1))*sin1}
Eci1 ci1 0 value={-(V(a1)*V(cr2)+V(wr)*V(b1))*sin1+
+(V(a1)*V(ci2)+V(wi)*V(b1))*cos1}
Ecr2 cr2 0 value={V(a2)*V(cr1)*cos2+
+V(a2)*V(ci1)*sin2}
Eci2 ci2 0 value={-V(a2)*V(cr1)*sin2+
+V(a2)*V(ci1)*cos2}
Emod1 mod1 0 value={sqrt(V(cr1)^2+V(ci1)^2)}
Epha1 pha1 0 value={atan2(V(ci1),V(cr1))*180/pi}
*****
.step dec param f 1k 3meg 50
.tran 0 0.9u skipbp
.probe
*****
.subckt SH1 in out
Ron in out {Ron}
C out 0 {C}
Rload out 0 {R}
.ends
*****
.subckt SH2 out
C out 0 {C}
Rload out 0 {R}
.ends
*****
.subckt TH in out
eTH out 0 value={if(time<=0.1u,v(in),state(0,V(out)))}
.ends
*****
.end
```




Fig. 3: Frequency responses, computed by numerical z-domain method (solid lines), compared to general s-z method (dashed lines), $f_s = 1\text{MHz}$.

The resulting frequency responses for a switching frequency of 1MHz, computed via the transient analysis, are shown in Fig. 3. The standard *YatLastX* measuring function is used in PROBE for evaluating the last value of circuit variable at the end of transient analysis. The dashed curves from the general s-z analysis are added for comparison. It should be noted that both results are nearly identical for the frequency range up to one half of switching frequency.

5 CONCLUSIONS

A novel method of AC analysis of linear circuits with externally and periodically controlled analog switches is described. This method utilizes the so-called generalized s-z transfer functions. In comparison with the classical methods based on averaged modeling, the advantages consist in a more faithful modeling of circuit behavior, particularly in the frequency range around $f_{\text{switch}}/2$, as well as in the ability to model correctly the transfers above this border frequency. A drawback consists in the necessity of analytical preprocessing of circuit models prior to their implementation in SPICE. An algorithm of automated building up of the corresponding circuit

matrices is published in [10]. However, it is incompatible with the internal SPICE algorithms. That is why it can be implemented only in special-purpose programs. To avoid this difficulty, an original method of AC analysis via conventional transient analysis in PSpice has been developed. This method is based on the assumption that the input signal remains unchanged during the switching phase. In spite of this simplification, the precision of AC analysis is little affected for frequencies up to one half of the switching frequency.

6 REFERENCES

- [1] Bedrosian, D. G., Vlach, J. An accelerated steady-state method for networks with internally controlled switches. *IEEE Trans. on Circuits and Systems I*, Vol. 39, No. 7, pp. 520- 530, 1992.
- [2] Valsa, J., Vlach, J. Swann - a Programme for Analysis of Switched Analogue Nonlinear Networks. *International Journal of Circuit Theory and Applications*, 1995, vol. 23, pp. 369-379.
- [3] Opal, A., Vlach, J. Analysis and sensitivity of periodically switched linear networks. *IEEE Trans. on Circuits and Systems I*, Vol. 36, No. 4, pp. 522 – 532, 1989.
- [4] Čajka, J., Dostál, T., Fencl, F. Computer-Aided Analysis of SC Circuits. In: *The Library of Scientific works*, VUT Brno, 1987, No. B-114, pp. 97-100 (In Czech).
- [5] Čajka, J. Computer-Aided Symbolic Analysis of Networks Containing Immittance Converters and Ideal Operational Amplifiers. *ISCAS'74*, pp. 515-519, IEEE, San Francisco, 1974.
- [6] Čajka, J., Dostál, T., Fencl, F. New Version of the COCO Program. *Radioengineering*, Vol. 3, No. 3, pp. 18-21, 1994.
- [7] Čajka, J., Vrba, K., Biolek, D. Improved program for symbolic analysis and synthesis of linear circuits. *AMSE'94*, Lyon, Vol.1, pp. 299-302, 1994, France.
- [8] Biolek, D. SCSK – program for analysis of switched capacitor circuits. *Electronic Horizon*, Vol. 50, No. 4, pp. 112 – 117, 1989 (In Czech).
- [9] Biolek, D. Computer analysis of waveforms in switched capacitor filters. *Electronic Horizon*, Vol. 51, 1990, No. 5, pp. 221 – 224 (In Czech).
- [10] Biolek, D. Modeling of Periodically Switched Networks by Mixed s-z Description. *IEEE Trans. on CAS-I*, Vol. 44, 1997, No. 8, pp. 750-758.
- [11] Opal, A., Vlach, J. Consistent initial conditions of linear switched networks. *IEEE Trans. On Circuits and Systems I*, Vol. 37, No. 3, pp. 364 – 372, 1990.
- [12] Biolek, D. Time domain analysis of linear systems using Laplace inversion. *Radioengineering*, No.1, 1994, pp.17-20.
- [13] Middlebrook, R.D., Čuk, S. A general unified approach to modeling switching-converter power stages. *Int. Journal of Electronics*, 1977, vol. 42, no. 6, pp. 512-550.
- [14] Basso, C.P. *Switch-Mode Power Supply SPICE Cookbook*. McGraw-Hill, 2001.
- [15] Biolek, D., Biolková, V., Dobeš, J. Modeling of switched DC-DC converters by mixed s-z description. In: *IEEE International Symposium on Circuits and Systems, ISCAS2006*, Greece, Kos, 2006, pp. 831 – 834.

Acknowledgment

This work is supported by the Grant Agency of the Czech Republic under grants No. 102/05/0771 and 102/05/0277, and by the research programmes of BUT MSM0021630503, MSM0021630513, and UD Brno MO FVT0000403.

The corresponding author:

Dalibor Biolek, Prof. Ing. CSc., UO Brno, K217,
Kounicova 65, 612 00 Brno, dalibor.biolek@unob.cz